

Chapter 16

Quantum Oscillator

Catching a lion, the Schrödinger's method: At any moment, there is a nonzero probability that a lion is inside the cage. Sit and wait.

16.1 Lowering and Raising Operators

The Hamiltonian of the harmonic oscillator is [18]

$$\hat{H} = \frac{\hat{p}^2}{2m} + \frac{m\omega^2 \hat{x}^2}{2}.$$

There is a dimensionful constant \hbar in quantum mechanics; therefore, two quantities m and ω define a scale of length $\sqrt{\hbar/(m\omega)}$, momentum $\sqrt{\hbar m\omega}$, energy $\hbar\omega$, and any other quantity of any dimensionality. They have the meaning of the characteristic amplitude, momentum, and energy of zero oscillations. We shall put $\hbar = 1$, $m = 1$, and $\omega = 1$, thus choosing these characteristic scales as units for measurement of corresponding quantities. Then

$$\hat{H} = \frac{\hat{p}^2 + \hat{x}^2}{2}.$$

Let's introduce the operator

$$\hat{a} = \frac{\hat{x} + i\hat{p}}{\sqrt{2}}$$

and its Hermitian conjugate

$$\hat{a}^+ = \frac{\hat{x} - i\hat{p}}{\sqrt{2}}.$$

The commutation relation $[\hat{p}, \hat{x}] = -i$ implies for them

$$[\hat{a}, \hat{a}^+] = 1.$$

The Hamiltonian is expressed via these operators as

$$\hat{H} = \hat{a}^+ \hat{a} + \frac{1}{2},$$

from where we obtain $[\hat{H}, \hat{a}] = -\hat{a}$, $[\hat{H}, \hat{a}^+] = \hat{a}^+$. Therefore, if $|\psi\rangle$ is an eigenstate of \hat{H} with the energy E : $\hat{H}|\psi\rangle = E|\psi\rangle$, then $\hat{a}|\psi\rangle$ and $\hat{a}^+|\psi\rangle$ are also eigenstates of \hat{H} with the energies $E - 1$ and $E + 1$:

$$\hat{H}\hat{a}|\psi\rangle = (\hat{a}\hat{H} - \hat{a})|\psi\rangle = (E - 1)\hat{a}|\psi\rangle, \quad \hat{H}\hat{a}^+|\psi\rangle = (E + 1)\hat{a}^+|\psi\rangle$$

(if only these states don't vanish). Hence the eigenvalues of \hat{H} form an arithmetic progression with step equal to 1. It is bounded from below because \hat{H} is a positive definite operator. Therefore, there exists a state $|0\rangle$ with the lowest energy that cannot be lowered any more:

$$\hat{a}|0\rangle = 0.$$

Its energy is equal to $\frac{1}{2}$:

$$\hat{H}|0\rangle = \left(\hat{a}^+ \hat{a} + \frac{1}{2}\right)|0\rangle = \frac{1}{2}|0\rangle$$

(this is the zero oscillations energy). Acting on $|0\rangle$ by the raising operator \hat{a}^+ n times, we obtain a state $|n\rangle$ with the energy

$$E_n = n + \frac{1}{2}.$$

Hence, $\hat{H}|n\rangle = (\hat{a}^+ \hat{a} + \frac{1}{2})|n\rangle = (n + \frac{1}{2})|n\rangle$ or

$$\hat{a}^+ \hat{a}|n\rangle = n|n\rangle,$$

i.e., $\hat{a}^+ \hat{a}$ acts as an operator of the level number.

We have $\hat{a}|n\rangle = c_n|n-1\rangle$; it is possible to make c_n real and positive by tuning the phases of the states $|n\rangle$. These coefficients can be found from the normalization condition: $|c_n|^2 = \langle n|\hat{a}^+ \hat{a}|n\rangle = n$. The action of the operator \hat{a}^+ follows from Hermitian conjugation:

$$\hat{a}|n\rangle = \sqrt{n}|n-1\rangle, \quad \hat{a}^+|n\rangle = \sqrt{n+1}|n+1\rangle.$$

From this we again have $\hat{a}^+ \hat{a}|n\rangle = n|n\rangle$.

In the coordinate representation

$$\hat{x} = x, \quad \hat{p} = -i \frac{d}{dx}.$$

Let's implement the operators \hat{a} and \hat{a}^+ in *Mathematica*.

In[1] := a[f_] := Together[(x * f + D[f, x])/Sqrt[2]]

In[2] := ac[f_] := Together[(x * f - D[f, x])/Sqrt[2]]

16.2 Ground State

This is the state which cannot be lowered by \hat{a} .

In[3] := Eq = a[ψ₀[x]] == 0

Out[3] = $\frac{x\psi_0[x] + \psi_0'[x]}{\sqrt{2}} == 0$

In[4] := s = DSolve[Eq, ψ₀[x], x]

Out[4] = $\left\{ \left\{ \psi_0[x] \rightarrow e^{-\frac{x^2}{2}} C[1] \right\} \right\}$

In[5] := ψ₀ = ψ₀[x]/.s[[1]]

Out[5] = $e^{-\frac{x^2}{2}} C[1]$

In[6] := Clear[Eq]

The normalization integral.

In[7] := NI = Integrate[ψ₀², {x, -Infinity, Infinity}]

Out[7] = $\sqrt{\pi} C[1]^2$

In[8] := s = Solve[NI == 1, C[1]]

Out[8] = $\left\{ \left\{ C[1] \rightarrow -\frac{1}{\pi^{1/4}} \right\}, \left\{ C[1] \rightarrow \frac{1}{\pi^{1/4}} \right\} \right\}$

In[9] := ψ₀ = ψ₀/.s[[2]]

Out[9] = $\frac{e^{-\frac{x^2}{2}}}{\pi^{1/4}}$

In[10] := Clear[NI, s]

16.3 Excited States

In[11] := ψ[0] = ψ₀;

In[12] := ψ[n_] := ψ[n] = ac[ψ[n - 1]]/Sqrt[n]

The wave functions of a few first states.


In[13] := Table[ψ[n], {n, 0, 10}]

Out[13] = $\left\{ \frac{e^{-\frac{x^2}{2}}}{\pi^{1/4}}, \frac{\sqrt{2}e^{-\frac{x^2}{2}}x}{\pi^{1/4}}, \frac{e^{-\frac{x^2}{2}}(-1+2x^2)}{\sqrt{2}\pi^{1/4}}, \frac{e^{-\frac{x^2}{2}}x(-3+2x^2)}{\sqrt{3}\pi^{1/4}}, \right.$
 $\frac{e^{-\frac{x^2}{2}}(3-12x^2+4x^4)}{2\sqrt{6}\pi^{1/4}}, \frac{e^{-\frac{x^2}{2}}x(15-20x^2+4x^4)}{2\sqrt{15}\pi^{1/4}},$
 $\frac{e^{-\frac{x^2}{2}}(-15+90x^2-60x^4+8x^6)}{12\sqrt{5}\pi^{1/4}}, \frac{e^{-\frac{x^2}{2}}x(-105+210x^2-84x^4+8x^6)}{6\sqrt{70}\pi^{1/4}},$
 $\left. \frac{e^{-\frac{x^2}{2}}(105-840x^2+840x^4-224x^6+16x^8)}{24\sqrt{70}\pi^{1/4}} \right\}$

$$\left. \begin{aligned} & \frac{e^{-\frac{x^2}{2}} x (945 - 2520x^2 + 1512x^4 - 288x^6 + 16x^8)}{72\sqrt{35}\pi^{1/4}}, \\ & \frac{e^{-\frac{x^2}{2}} (-945 + 9450x^2 - 12600x^4 + 5040x^6 - 720x^8 + 32x^{10})}{720\sqrt{7}\pi^{1/4}} \end{aligned} \right\}$$

And here the level number can be set by the mouse.

In[14] := Manipulate[$\psi[n]$, { n , 0, 10, 1, Appearance -> "Labeled"}]




Out[14] =

$$\frac{e^{-\frac{x^2}{2}} x (-3 + 2x^2)}{\sqrt{3}\pi^{1/4}}$$

The wave functions of a few first states.


In[15] := Plot[Evaluate[Table[$\psi[n]$, { n , 0, 3}], { x , -5, 5}]



Out[15] =

And this is a live plot: the level number can be set by the mouse.

In[16] := Manipulate[Plot[$\psi[n]$, { x , -5, 5}, PlotRange -> {-0.8, 0.8}], { n , 0, 10, 1, Appearance -> "Labeled"}]



16.4 Some Properties

Orthogonality and normalization.

```
In[17] := Distribute[ψ]
```

Out[17] = ψ

```
In[18] := Parallelize[Table[Table[Integrate[ψ[n] * ψ[m], {x, -Infinity, Infinity}],
    {n, 0, 10}], {m, 0, 10}]]
```

```
Out[18] = {{1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0},
    {0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0},
    {0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0},
    {0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0},
    {0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0},
    {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1}}
```

Wave functions in the momentum representation (Fourier transforms) are the same as in the coordinate one, up to phase factors.


```
In[19] := Parallelize[Table[
    Cancel[Integrate[ψ[n] * Exp[-I * p * x], {x, -Infinity, Infinity}]] /
    Sqrt[2 * Pi] / (ψ[n] /. x -> p),
    {n, 0, 10}]]
```

```
Out[19] = {1, -i, -1, i, 1, -i, -1, i, 1, -i, -1}
```

The probability density.

```
In[20] := Manipulate[Plot[ψ[n]^2, {x, -6, 6}, PlotRange -> {0, 0.6}],
    {{n, 10}, 0, 10, 1, Appearance -> "Labeled"}]
```

Out[20] =



Why is it larger near the boundaries?